
MS5607 Calibration

This document describes the calibration method devised by David Edwards for performing a first 
order calibration of the MS5607-02BA03 Barometric Pressure Sensor on an Arduino processor, 
which is limited to 32 bit integer operations.

The MS5607 contains factory calibration constants, which must be used to calibrate the raw 
temperature and pressure data, as described in the MS5607 specifications:

FACTORY CALIBRATION 

Every  module  is  individually  factory  calibrated  at  two  temperatures  and  two  pressures.  As  a  result,  6 
coefficients necessary to compensate for process variations and temperature variations are calculated and 
stored in the 128-bit PROM of each module. These bits (partitioned into 6 coefficients) must be read by the 
microcontroller software and used in the program converting D1 and D2 into compensated pressure and 
temperature values. 

The calculations involve operations on signed and unsigned integers with a maximum size of 64 
bits. The standard Arduino Long integers have only 32 bit capacity, so if the algorithm specified in 
Fig 2 above is used, the calculations are inaccurate because of over and under flows.

This is because the algorithm requires scaling by factors of up to 2^23.

The first order calibration uses the following algorithm:

P = (D1 x SENS / 2097152  – OFF) / 65536

where:

P = Pressure in Pa
D1 is the raw pressure data
SENS and OFF are the calculated sensitivity (gain) and offset 

calculated from the calibration coefficients and the raw 
temperature reading.

SENS and OFF are calculated as follows:

SENS = C1 x 65536  + C3 x DT /128

OFF = C2 x 131072 + C4 x DT / 64

where:

C1, C2, C3 & C4 are factory calibration coefficients read from the MS5607 
sensor

DT is the difference difference

DT is calculated as follows:

DT = D2 – C5 x 256
where:

D2 is the raw temperature data
C5 is a factory calibration coefficient read from the MS5607 sensor

The MS5607 temperature is not needed to calculate pressure, but can be calculated as follows:

TEMP = 2000 + DT x C6 / 8388608

where:

TEMP is the calibrated temperature in degC, multiplied by 100
DT is the temperature difference
C6 is a factory calibration coefficient read from the MS5607 sensor

By algebraically expanding these equations, pressure P can be calculated as:

P = P1 + P2 – P3 – P4

where:

P1 = (D1 / 1024) x C1 / 1024)

P2 = (D1 / 16384) x (C3 / 16384) x (DT / 32768)

P3 = C2 x 4

P4 = (C4 / 1024) x (DT / 2048)

Each of the bracketed terms is evaluated before the terms are multiplied.

To obtain air pressure in hPa or mBars, a further division by 100 is performed.

Arduino code to perform these calculations is shown below. Note that nSamples are taken, 
calibrated and then averaged:

  // Read pressure and temperature
  TempAccum = 0;
  PressAccum = 0;
  for(i=0; i<nSamples; i++)
  {
    // read temperature
    RawTemp = ReadADC(0x18);   // typical value 8077636 
    DeltaTemp = RawTemp - coefficients_[4] * 256;
    CalTemp = 2000 + (DeltaTemp * coefficients_[5] >> 23);
    TempAccum += CalTemp;
    // read pressure
    RawPress = ReadADC(0x08);  // typical value 6465444
    P1 = (float(RawPress)/1024L)*(float(coefficients_[0])/1024L);
    P2 = (float(RawPress)/16384L)*(float(coefficients_[2])/16384L)*(float(DeltaTemp)/32768L);
    P3 = (float(coefficients_[1])*4);
    P4 = (float(coefficients_[3])/1024l)*(float(DeltaTemp)/2048L);
    PressAccum += P1 + P2 - P3 - P4;
  }
  Temp = TempAccum/nSamples/100;
  Press = PressAccum/nSamples/100;

Arduino code to read the calibration coefficients is shown below:

    unsigned int ReadCoefficient(const byte coefNum)
    {
        unsigned int rC=0;
        Wire.beginTransmission(i2cAddr_);
        Wire.write(cmdPromRd_ + coefNum * 2); // send PROM READ command
        Wire.endTransmission(); 
        const byte q2 = 2;
        Wire.requestFrom(i2cAddr_, q2);
        if(Wire.available() >= 2)
        {
            unsigned int ret = Wire.read();   // read MSB
            unsigned int rC  = 256 * ret;
            ret = Wire.read();                // read LSB
            rC  = rC + ret;
            return rC;
        } 
        return 0;
    }

Note that Cofficient C1 is stored in the coefficients_[0] array element.

The spreadsheet extract below shows a typical calibration:

The upper block shows the temperature and pressure calculations performed using the MS5607 
calculation method, which gives the correct answers because of the high precision spreadsheet 
calculations.

The lower block shows the pressure calculations performed as per the method outlined above.

David Edwards
8 August 2014

